NAG Fortran Library Routine Document

F07PDF (SSPTRF/DSPTRF)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07PDF (SSPTRF/DSPTRF) computes the Bunch–Kaufman factorization of a real symmetric indefinite matrix, using packed storage.

2 Specification

```
SUBROUTINE F07PDF(UPLO, N, AP, IPIV, INFO)
ENTRY      ssptrf  (UPLO, N, AP, IPIV, INFO)
INTEGER            N, IPIV(*), INFO
real               AP(*)
CHARACTER*1        UPLO
```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine factorizes a real symmetric matrix A, using the Bunch–Kaufman diagonal pivoting method and packed storage. A is factorized as either $A = PUDU^TP^T$ if UPLO = 'U', or $A = PLDL^TP^T$ if UPLO = 'L', where P is a permutation matrix, U (or L) is a unit upper (or lower) triangular matrix and D is a symmetric block diagonal matrix with 1 by 1 and 2 by 2 diagonal blocks; U (or L) has 2 by 2 unit diagonal blocks corresponding to the 2 by 2 blocks of D. Row and column interchanges are performed to ensure numerical stability while preserving symmetry.

This method is suitable for symmetric matrices which are not known to be positive-definite. If A is in fact positive-definite, no interchanges are performed and no 2 by 2 blocks occur in D.

4 References

Golub G H and van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: UPLO – CHARACTER*1 *Input*

On entry: indicates whether the upper or lower triangular part of A is stored and how A is to be factorized, as follows:

if UPLO = 'U', the upper triangular part of A is stored and A is factorized as $PUDU^TP^T$, where U is upper triangular;

if UPLO = 'L', the lower triangular part of A is stored and A is factorized as $PLDL^TP^T$, where L is lower triangular.

Constraint: UPLO = 'U' or 'L'.

2: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N $\geq$ 0.

3: AP(∗) – ***real*** array *Input/Output*

Note: the dimension of the array AP must be at least $\max(1, N * (N+1)/2)$.

On entry: the n by n symmetric matrix A, packed by columns. More precisely, if UPLO = 'U', the upper triangle of A must be stored with element a_{ij} in $AP(i + j(j-1)/2)$ for $i \leq j$; if UPLO = 'L', the lower triangle of A must be stored with element a_{ij} in $AP(i + (2n-j)(j-1)/2)$ for $i \geq j$.

On exit: A is overwritten by details of the block diagonal matrix D and the multipliers used to obtain the factor U or L as specified by UPLO.

4: IPIV(∗) – INTEGER array *Output*

Note: the dimension of the array IPIV must be at least $\max(1, N)$.

On exit: details of the interchanges and the block structure of D. More precisely, if $\text{IPIV}(i) = k > 0$, then d_{ii} is a 1 by 1 pivot block and the ith row and column of A was interchanged with the kth row and column. If UPLO='U' and $\text{IPIV}(i-1) = \text{IPIV}(i) = -l < 0$, then $\begin{pmatrix} d_{i-1,i-1} & d_{i,i-1} \\ d_{i,i-1} & d_{ii} \end{pmatrix}$ is a 2 by 2 pivot block and the $(i-1)$th row and column of A was interchanged with the lth row and column; if UPLO = 'L' and $\text{IPIV}(i) = \text{IPIV}(i+1) = -m < 0$, then $\begin{pmatrix} d_{ii} & d_{i+1,i} \\ d_{i+1,i} & d_{i+1,i+1} \end{pmatrix}$ is a 2 by 2 pivot block and the $(i+1)$th row and column of A was interchanged with the mth row and column.

5: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO $= i$, d_{ii} is exactly zero. The factorization has been completed but the block diagonal matrix D is exactly singular, and division by zero will occur if it is subsequently used to solve a system of linear equations or to compute A^{-1}.

7 Accuracy

If UPLO = 'U', the computed factors U and D are the exact factors of a perturbed matrix $A + E$, where

$$|E| \leq c(n)\epsilon P|U|\,|D|\,|U^T|P^T,$$

$c(n)$ is a modest linear function of n, and ϵ is the ***machine precision***.

If UPLO = 'L', a similar statement holds for the computed factors L and D.

8 Further Comments

The elements of D overwrite the corresponding elements of A; if D has 2 by 2 blocks, only the upper or lower triangle is stored, as specified by UPLO.

The unit diagonal elements of U or L and the 2 by 2 unit diagonal blocks are not stored. The remaining elements of U or L overwrite elements in the corresponding columns of A, but additional row interchanges must be applied to recover U or L explicitly (this is seldom necessary). If $\text{IPIV}(i) = i$, for $i = 1, 2, \ldots, n$

(as is the case when A is positive-definite), then U or L are stored explicitly in packed form (except for their unit diagonal elements which are equal to 1).

The total number of floating-point operations is approximately $\frac{1}{3}n^3$.

A call to this routine may be followed by calls to the routines:

F07PEF (SSPTRS/DSPTRS) to solve $AX = B$;

F07PGF (SSPCON/DSPCON) to estimate the condition number of A;

F07PJF (SSPTRI/DSPTRI) to compute the inverse of A.

The complex analogues of this routine are F07PRF (CHPTRF/ZHPTRF) for Hermitian matrices and F07QRF (CSPTRF/ZSPTRF) for symmetric matrices.

9 Example

To compute the Bunch–Kaufman factorization of the matrix A, where

$$A = \begin{pmatrix} 2.07 & 3.87 & 4.20 & -1.15 \\ 3.87 & -0.21 & 1.87 & 0.63 \\ 4.20 & 1.87 & 1.15 & 2.06 \\ -1.15 & 0.63 & 2.06 & -1.81 \end{pmatrix},$$

using packed storage.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F07PDF Example Program Text
*     Mark 15 Release. NAG Copyright 1991.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX
      PARAMETER        (NMAX=8)
*     .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, J, N
      CHARACTER        UPLO
*     .. Local Arrays ..
      real             AP(NMAX*(NMAX+1)/2)
      INTEGER          IPIV(NMAX)
*     .. External Subroutines ..
      EXTERNAL         ssptrf, X04CCF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F07PDF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*        Read A from data file
*
         READ (NIN,*) UPLO
         IF (UPLO.EQ.'U') THEN
            READ (NIN,*) ((AP(I+J*(J-1)/2),J=I,N),I=1,N)
         ELSE IF (UPLO.EQ.'L') THEN
            READ (NIN,*) ((AP(I+(2*N-J)*(J-1)/2),J=1,I),I=1,N)
         END IF
*
*        Factorize A
*
         CALL ssptrf(UPLO,N,AP,IPIV,INFO)
*
         WRITE (NOUT,*)
```

```
*
*        Print details of factorization
*
         IFAIL = 0
*
         CALL X04CCF(UPLO,'Nonunit',N,AP,'Details of factorization',
     +               IFAIL)
*
*        Print pivot indices
*
         WRITE (NOUT,*)
         WRITE (NOUT,*) 'IPIV'
         WRITE (NOUT,99999) (IPIV(I),I=1,N)
*
         IF (INFO.NE.0) WRITE (NOUT,*) 'The factor D is singular'
*
      END IF
      STOP
*
99999 FORMAT ((3X,7I11))
      END
```

9.2 Program Data

```
F07PDF Example Program Data
  4                           :Value of N
  'L'                         :Value of UPLO
  2.07
  3.87  -0.21
  4.20   1.87   1.15
 -1.15   0.63   2.06  -1.81   :End of matrix A
```

9.3 Program Results

```
 F07PDF Example Program Results

 Details of factorization
             1          2          3          4
 1      2.0700
 2      4.2000     1.1500
 3      0.2230     0.8115    -2.5907
 4      0.6537    -0.5960     0.3031     0.4074

 IPIV
           -3         -3          3          4
```